
Nov. 23 2006

Lecture 23: Resonance Phenomena, Beam Theory

Reading:

Kreyszig Sections: 2.8, 2.9, 3.1, 3.2, 3.3 (pages 84–90, 91–96, 105–111, 111–115, 116–121)

Resonance Phenomena

The physics of an isolated damped linear harmonic oscillator follows from the behavior of the homogeneous equation:¹⁵

$$M \frac{d^2 y(t)}{dt^2} + \eta l_o \frac{dy(t)}{dt} + K_s y(t) = 0 \quad (23-1)$$

This equation is the sum of three forces:

inertial force depending on the acceleration of the object.

drag force depending on the velocity of the object.

spring force depends on the displacement of the object.

The system is *autonomous* in the sense that everything depends on the system itself; there are no outside agents changing the system.

The zero on the right-hand-side of Eq. 23-1 implies that there are no external forces applied to the system. The system oscillates with a characteristic frequency $\omega = \sqrt{K_s/M}$ with amplitude that are damped by a characteristic time $\tau = (2M)/(\eta l_o)$ (i.e., the amplitude is damped $\propto \exp(-t/\tau)$.)

¹⁵ A concise and descriptive description of fairly general harmonic oscillator behavior appears at <http://hypertextbook.com/chaos/41.shtml>

Lecture 23 MATHEMATICA® Example 1

Simulating Harmonic Oscillation with Biased and Unbiased Noise

Download notebooks, pdfs, or html from <http://pruffle.mit.edu/3.016-2006>.

The second-order differencing simulation of a harmonic oscillator is modified to include white and biased stochastic nudging.

- 1: *GrowListGeneralNoise* is extended from a previous example for simulating $\ddot{y} + \beta\dot{y} + \alpha y = 0$ (*GrowList* in example 21-1) and adds a random uniform displacement $y + \delta$, $\delta \in (-\text{randomamp}, \text{randomamp})$ at each iteration. The *ValuesList_List* argument should be a list containing two lists: the first list is comprised of the sequence of displacements y ; the second list records the corresponding stochastic displacement δ . The function uses a list's two previous values and **Append** and to grow the list iteratively.
- 4: Exemplary data from 2×10^5 iterations (using **Nest**) is produced for the specific case of $\Delta = 0.001$, $\alpha = 2$, $\beta = 0$.
- 5: The displacements (i.e., first list) are plotted with **ListPlot**.
- 6: The random 'nudges' (i.e., second list) are also plotted.
- 7: Biased nudges are simulated with *GrowListBiasedNoise*. This extends the unbiased example above, by including a wavelength for a cosine-biased random amplitude. A sample, δ , from the uniform random distribution as above is selected and then multiplied by $\cos 2\pi t/\lambda$. The time-like variable is simulated with **Length** and the current data.
- 10: The biased data for approximately the resonance condition for the same model parameters above is plotted with the biased noise.

```

GrowListGeneralNoise[ValuesList_List, Δ_, α_, β_,
  randomamp_] := Module[{Minus1 = ValuesList[[1, -1]],
    Minus2 = ValuesList[[1, -2]],
    noise = Random[Real, {-randomamp, randomamp}]},
  {Append[ValuesList[[1]],
    2 * Minus1 - Minus2 + Δ *
      (β * (Minus2 - Minus1) - α * Δ * Minus2) + noise],
    Append[ValuesList[[2]], noise]}]

GrowListSpecificNoise[InitialList_List] :=
  GrowListGeneralNoise[InitialList, .001, 2, 0, 10^(-5)]

Nest[GrowListSpecificNoise, {{1, 1}, {0, 0}}, 10]

TheData =
  Nest[GrowListSpecificNoise, {{1, 1}, {0, 0}}, 20000];

ListPlot[TheData[[1]]]

ListPlot[TheData[[2]]]

Now suppose there is a periodic bias that tends to kick the
displacement one direction more than the other:

GrowListBiasedNoise[ValuesList_List,
  Δ_, α_, β_, randomamp_, lambda_] :=
  Module[{Minus1 = ValuesList[[1, -1]], Minus2 =
    ValuesList[[1, -2]], biasednoise = 0.5 * randomamp *
      (Cos[2 * Length[ValuesList[[1]]] / lambda] +
        Random[Real, {-1, 1}])},
  {Append[ValuesList[[1]],
    2 * Minus1 - Minus2 +
      Δ * (β * (Minus2 - Minus1) - α * Δ * Minus2) +
        biasednoise],
    Append[ValuesList[[2]], biasednoise]}]

GrowListSpecificBiasedNoise[InitialList_List] :=
  GrowListBiasedNoise[InitialList, .001, 2, 0, 10^(-6), 4500]

TheBiasedData =
  Nest[GrowListSpecificBiasedNoise, {{1, 1}, {0, 0}}, 20000];

ListPlot[TheBiasedData[[1]]]
ListPlot[TheBiasedData[[2]]]

```

A general model for a damped and forced harmonic oscillator is

$$M \frac{d^2 y(t)}{dt^2} + \eta l_o \frac{dy(t)}{dt} + K_s y(t) = F_{app}(t) \quad (23-2)$$

where F_{app} represents a time-dependent applied force to the mass M .

General Solutions to Non-homogeneous ODEs

Equation 23-2 is a non-homogeneous ODE—the functions and its derivatives appear on one side and an arbitrary function appears on the other. The general solution to Eq. 23-2 will be the sum of two parts:

$$\begin{aligned} y_{gen}(t) &= y_{part}(t) + y_{homog}(t) \\ y_{gen}(t) &= y_{F_{app}}(t) + y_{homog}(t) \end{aligned} \quad (23-3)$$

$$y_{homog}(t) = \begin{cases} C_+ e^{-|\lambda|t} + C_- e^{-|\lambda|t} & (\eta l_o)^2 > 4MK_s \text{ Over-damped} \\ C_1 e^{-|\lambda|t} + C_2 t e^{-|\lambda|t} & (\eta l_o)^2 = 4MK_s \text{ Critical Damping} \\ C_+ e^{-|\text{Re}\lambda|t} e^{i|\text{Im}\lambda|t} + C_- e^{-|\text{Re}\lambda|t} e^{-i|\text{Im}\lambda|t} & (\eta l_o)^2 < 4MK_s \text{ Under-damped} \end{cases} \quad (23-4)$$

where $y_{part} \equiv y_{F_{app}}$ is the solution for the particular F_{app} on the right-hand-side and y_{homog} is the solution for the right-hand-side being zero. *Adding the homogeneous solution y_{homog} to the particular solution y_{part} is equivalent to adding a “zero” to the applied force F_{app}*

Interesting cases arise when the applied force is periodic $F_{app}(t) = F_{app}(t+T) = F_{app}(t+2\pi/\omega_{app})$, especially when the applied frequency, ω_{app} is close to the the characteristic frequency of the oscillator $\omega_{char} = \sqrt{K_s/M}$.

Modal Analysis

For the case of a periodic forcing function, the time-dependent force can be represented by a Fourier Series. Because the second-order ODE (Eq. 23-2) is linear, the particular solutions for each term in a Fourier series can be summed. Therefore, particular solutions can be analyzed for one trigonometric term at a time:

$$M \frac{d^2 y(t)}{dt^2} + \eta l_o \frac{dy(t)}{dt} + K_s y(t) = F_{app} \cos(\omega_{app} t) \quad (23-5)$$

There are three general cases for the particular solution:

	Condition	Solution for $F(t) = F_{app} \cos(\omega_{app} t)$
Undamped, Frequency- Mismatch	$\eta = 0$ $\omega_{char}^2 = \frac{K_s}{M} \neq \omega_{app}^2$	$y_{part}(t) = \frac{F_{app} \cos(\omega_{app} t)}{M(\omega_{char} + \omega_{app})(\omega_{char} - \omega_{app})}$
Undamped, Frequency- Matched	$\eta = 0$ $\omega_{char}^2 = \frac{K_s}{M} = \omega_{app}^2$	$y_{part}(t) = \frac{F_{app} t \sin(\omega_{app} t)}{2M\omega_{app}}$
Damped	$\eta > 0$	$y_{part}(t) = \frac{F_{app} \cos(\omega_{app} t + \phi_{lag})}{\sqrt{M^2(\omega_{char}^2 - \omega_{app}^2)^2 + \omega_{app}^2 \eta^2 l_o^2}}$ $\phi_{lag} = \tan^{-1} \left(\frac{\omega_{app} \eta l_o}{M(\omega_{char}^2 - \omega_{app}^2)} \right)$

The phenomenon of resonance can be observed as the driving frequency approaches the characteristic frequency.

Lecture 23 MATHEMATICA® Example 2

Resonance and Near-Resonance Behavior

Download notebooks, pdfs, or html from <http://pruffle.mit.edu/3.016-2006>.

Solutions to $m\ddot{y} + \eta\dot{y} + ky = F_{app} \cos(\omega_{app}t)$ analyzed near the resonance condition $\omega_{app} \approx \omega_{char} \equiv \sqrt{k/m}$.

- 2: The general solution will include two arbitrary constants `C[1]` and `C[2]` in terms that derive from the homogeneous solution plus a part that derives from the heterogeneous (i.e., forced) part.
- 3: Examining the form of the general solution at $t = 0$, it will be clear that the constants from the homogeneous part will be needed to satisfy arbitrary boundary conditions—most importantly, the constants will include terms that depend on the characteristic and applied frequencies.
- 4: Here `DSolve` will be used *yParticularSolution* to analyze the particular case of a forced ($F(t) = F_{app} \cos(\omega_{app}t)$) and damped harmonic oscillator initially at resting equilibrium ($y(t = 0) = 1$ and $y'(t = 0) = 0$).
- 5: The most interesting cases are the resonance and near resonance cases: *ResonantSolution* is obtained by setting the forcing frequency equal to the characteristic frequency.
- 6: To analyze the at-resonance case, the solution will be expanded to second order for small viscosity with `Series`. Some extra manipulation is required to display the results in a form that is straightforward to interpret. Here, `Map` will be used with a *pure function* to simplify each term produced by `Series`. First, the `SeriesData` object created by `Series` is transformed into a regular expression with `Normal`. The pure function will first transform any $\exp(x)$ into $\cosh(x) + \sinh(x)$, then any fractional powers will be cleaned up (e.g., $\sqrt{x^2} \rightarrow x$) assuming real parameters; finally the individual terms will be simplified.
- 7: This illustrates how near resonance $\omega_{app} \approx \omega_{char}$ can be analyzed in the small viscosity limit. Here, `Series` first expands around $\eta = 0$ to second order and then around small $\delta\omega = \omega_{app} - \omega_{char}$.
- 8: Setting the viscosity to zero *a priori* is possible and returns the leading order behavior, but the *asymptotic behavior* for small parameters cannot be ascertained.

Apply a forcing function: $F_{app} \cos(\omega_{app}t)$
To solve problems in terms of the mass and natural frequency, eliminate the spring constant in equations by defining it in terms of the mass and natural frequency.

```
1 Kspring = M ωchar2
```

Mathematica can solve the nonhomogeneous ODE with a forcing function at with an applied frequency:

```
2 yGeneralSol = Simplify[y[t] /. DSolve[{M y''[t] + η y'[t] + Kspring y[t] = Fapp Cos[ωapp t], y[0] == 0, y'[0] == 0}, y[t], t][[1]]]
```

Consider the behavior of the general solution at time $t=0$. This will show that the homogeneous parts of the solution are needed to satisfy boundary conditions, even if the oscillator is initially at rest at zero displacement (i.e., $y(0) = \dot{y}(0) = 0$).

```
3 Simplify[yGeneralSol /. t -> 0]
```

Consider the particular case of an equilibrium at-rest oscillator

```
4 yParticularSol = Simplify[y[t] /. DSolve[{M y''[t] + η y'[t] + Kspring y[t] = Fapp Cos[ωapp t], y[0] == 0, y'[0] == 0}, y[t], t][[1]]]
```

The resonant solution is the case: $\omega_{app} \rightarrow \omega_{char}$

```
5 ResonantSolution = Simplify[yParticularSol /. ωapp -> ωchar]
```

```
6 ResonantSolutionSmallViscosity = Map[Simplify[PowerExpand[ExpToTrig[#]]] &, Normal[Series[ResonantSolution, {η, 0, 2}]]]
```

```
7 ResonantSolutionSmallViscosityDetuned = Map[Simplify[PowerExpand[ExpToTrig[#]]] &, Normal[Series[yParticularSol, {ωapp, ωchar, 1}, {η, 0, 2}]]]
```

The leading behavior could have been obtained directly, viz

```
8 ResonantSolZeroViscosity = Simplify[y[t] /. DSolve[{M y''[t] + Kspring y[t] = Fapp Cos[ωchar t], y[0] == 0, y'[0] == 0}, y[t], t][[1]]]
```

Lecture 23 MATHEMATICA® Example 3

Visualizing Forced and Damped Harmonic Oscillation

Download notebooks, pdfs, or html from <http://pruffle.mit.edu/3.016-2006>.

- 1: This function solves the heterogeneous damped harmonic oscillator ODE (where $F(t) = \cos(\omega_{app}t)$) for any input mass, damping coefficient, and spring constant M , η , $k = M\omega_{char}^2$.
- 2: Undamped resonance $\omega_{char} = \omega_{app} = 1/2$ should show linearly growing amplitude.
- 3: Near resonance will show a beat-phenomena because of "de-tuning."
- 4: Damped resonance will show that the amplitudes approaching to a finite asymptotic limit.
- 6: The beats will still be apparent for the damped near resonance condition, but the finite damping coefficient will prevent the amplitude from completely disappearing.

Create a <i>Mathematica</i> function that returns the solution for specified mass, viscous term, characteristic and applied frequencies	
1	<code>y[M_, η_, ωchar_, ωapp_] := Chop[y[t] /. DSolve[{M y''[t] + η y'[t] + M ωchar^2 y[t] == Cos[ωapp t], y[0] == 1, y'[0] == 0}, y[t], t] // Flatten]</code>
Undamped Resonance:	
2	<code>Plot[Evaluate[y[1, 0, 1/2, 1/2]], {t, 0, 200}, PlotPoints -> 200]</code>
Undamped Near Resonance:	
3	<code>Plot[Evaluate[y[1, 0, 1/2 + 0.05, 1/2]], {t, 0, 200}, PlotPoints -> 200]</code>
Damped Resonance:	
4	<code>Plot[Evaluate[y[1, 1/10, 1/2, 1/2]], {t, 0, 200}]</code>
Overdamped Resonance:	
5	<code>Plot[Evaluate[y[1, 10, 1/2, 1/2]], {t, 0, 200}]</code>
Damped Near Resonance:	
6	<code>Plot[Evaluate[y[1, .05, 1/2 + 0.05, 1/2]], {t, 0, 200}, PlotPoints -> 200]</code>
Heavily damped Near Resonance:	
7	<code>Plot[Evaluate[y[1, 2.5, 1/2 + 0.05, 1/2]], {t, 0, 200}, PlotPoints -> 200]</code>

Resonance can have catastrophic or amusing (or both) consequences:




Figure 23-24: Picture and illustration of the bells at Kendall square. Many people shake the handles vigorously but with apparently no pleasant effect. The concept of resonance can be used to to operate the bells efficiently Perturb the handle slightly and observe the frequencies of the the pendulums—select one and wiggle the handle at the pendulum's characteristic frequency. The amplitude of that pendulum will increase and eventually strike the neighboring tubular bells.

From Cambridge Arts Council Website:

http://www.ci.cambridge.ma.us/~CAC/public_art_tour/map_11_kendall.html

Artist: Paul Matisse

Title: The Kendall Band - Kepler, Pythagoras, Galileo

Date: 1987

Materials: Aluminum, teak, steel

Handles located on the platforms allow passengers to play these mobile-like instruments, which are suspended in arches between the tracks, "Kepler" is an aluminum ring that will hum for five minutes after it is struck by the large teak hammer above it. "Pythagoras" consists of a 48-foot row of chimes made from heavy aluminum tubes interspersed with 14 teak hammers. "Galileo" is a large sheet of metal that rattles thunderously when one shakes the handle.

